Integrazione di funzioni razionali fratte

Autore: Claudio Marsan
Ultima revisione: 16 marzo 2003
Versione: Maple V Release 6.02 for Windows 2000

```
> restart:
```

In questo foglio di lavoro vogliamo mostrare come si integra una funzione razionale fratta passo per passo usando Maple.

La funzione razionale fratta da integrare:

```
> F := (4*x^6 + 30*x^3 + 59)/(2*x^5 + 3*x^4 - 10*x^3 - 17*x^2 -
  12*x - 20);
```

$$F := \frac{4\,x^6 + 30\,x^3 + 59}{2\,x^5 + 3\,x^4 - 10\,x^3 - 17\,x^2 - 12\,x - 20}$$

Numeratore e denominatore della funzione da integrare:

```
> num := numer(F); den := denom(F);
```

$$num := 4\,x^6 + 30\,x^3 + 59$$

$$den := 2\,x^5 + 3\,x^4 - 10\,x^3 - 17\,x^2 - 12\,x - 20$$

L'integrale da calcolare:

```
> Int(F, x);
```

$$\int \frac{4\,x^6 + 30\,x^3 + 59}{2\,x^5 + 3\,x^4 - 10\,x^3 - 17\,x^2 - 12\,x - 20}\,dx$$

Visto che il grado del numeratore è maggiore del grado del denominatore eseguiamo la divisione fra i due polinomi ottenendo un quoziente ed un resto:

```
> q := sort(quo(num, den, x)); r := sort(rem(num, den, x));
```

$$q := 2\,x - 3$$

$$r := 29\,x^4 + 34\,x^3 - 27\,x^2 + 4\,x - 1$$

Il problema iniziale diventa allora:

```
> Int(F, x) = Int(q, x) + Int(r/den, x);
```

$$\int \frac{4\,x^6 + 30\,x^3 + 59}{2\,x^5 + 3\,x^4 - 10\,x^3 - 17\,x^2 - 12\,x - 20}\,dx =$$
$$\int 2\,x - 3\,dx + \int \frac{29\,x^4 + 34\,x^3 - 27\,x^2 + 4\,x - 1}{2\,x^5 + 3\,x^4 - 10\,x^3 - 17\,x^2 - 12\,x - 20}\,dx$$

Il primo integrale è semplice, il secondo richiede un po' di lavoro. Cominciamo fattorizzando il denominatore:

```
> den := factor(den);
```

$$den := (2\,x - 5)\,(x^2 + 1)\,(x + 2)^2$$

L'integrale da calcolare è dunque:

```
> Int(r/den, x);
```

$$\int \frac{29\,x^4+34\,x^3-27\,x^2+4\,x-1}{(2\,x-5)\,(x^2+1)\,(x+2)^2}\,dx$$

Ora bisogna scrivere la funzione razionale fratta come una somma di funzioni razionali fratte "più semplici":

```
> desiderato := a/(2*x-5) + (b+c*x)/(x^2+1) + d/(x+2) + e/(x+2)^2;
```

$$desiderato := \frac{a}{2\,x-5}+\frac{b+c\,x}{x^2+1}+\frac{d}{x+2}+\frac{e}{(x+2)^2}$$

Dobbiamo ora determinare i coefficienti a, b, c, d, e. Dobbiamo dunque costruire un sistema di 5 equazioni lineari in 5 incognite. Dapprima abbiamo:

```
> r/den - desiderato = 0;
```

$$\frac{29\,x^4+34\,x^3-27\,x^2+4\,x-1}{(2\,x-5)\,(x^2+1)\,(x+2)^2}-\frac{a}{2\,x-5}-\frac{b+c\,x}{x^2+1}-\frac{d}{x+2}-\frac{e}{(x+2)^2}=0$$

Semplificando il membro a sinistra si ottiene:

```
> normal(lhs(%));
```

$$-(1-34\,x^3-29\,x^4+27\,x^2-4\,x+4\,a-20\,b-5\,e-10\,d-20\,c\,x+a\,x^4+4\,a\,x^3+5\,a\,x^2 +4\,a\,x-12\,b\,x-12\,c\,x^2+2\,b\,x^3+3\,b\,x^2+2\,c\,x^4+3\,c\,x^3-d\,x^3-8\,d\,x^2-d\,x+2\,e\,x^3 -5\,e\,x^2+2\,e\,x+2\,d\,x^4)\Big/((2\,x-5)\,(x^2+1)\,(x+2)^2)$$

A questo punto ci interessa unicamente il numeratore, che deve essere 0. Mettiamo in evidenza anche la variabile x:

```
> collect(numer(%),x);
```

$$(29-a-2\,c-2\,d)\,x^4+(-2\,e-3\,c-2\,b+34+d-4\,a)\,x^3 +(-27+12\,c+5\,e-3\,b+8\,d-5\,a)\,x^2+(12\,b+d+4+20\,c-2\,e-4\,a)\,x-1-4\,a+20\,b +5\,e+10\,d$$

Il polinomio sopra deve essere il polinomio nullo, e dunque i suoi coefficienti devono essere tutti 0. Si ottiene così il sistema di 5 equazioni lineari in 5 incognite:

```
> equazioni := {coeffs(%,x)};
```

$$equazioni := \{29-a-2\,c-2\,d, -2\,e-3\,c-2\,b+34+d-4\,a, -1-4\,a+20\,b+5\,e+10\,d, 12\,b+d+4+20\,c-2\,e-4\,a, -27+12\,c+5\,e-3\,b+8\,d-5\,a\}$$

Risolviamo ora il sistema:

```
> soluzione := solve(equazioni);
```

$$soluzione := \left\{d=\frac{994}{135}, e=\frac{-5}{3}, c=\frac{292}{145}, b=\frac{-169}{145}, a=\frac{8023}{783}\right\}$$

Sostituendo i coefficienti si ha la scomposizione in frazione parziale:

```
> frazione_parziale := subs(soluzione, desiderato);
```

$$frazione\_parziale := \frac{8023}{783}\,\frac{1}{2\,x-5}+\frac{-\frac{169}{145}+\frac{292}{145}\,x}{x^2+1}+\frac{\frac{994}{135}}{x+2}-\frac{5}{3}\,\frac{1}{(x+2)^2}$$

L'integrale da calcolare diventa dunque:

```
> map(Int, frazione_parziale, x);
```

$$\int \frac{8023}{783}\,\frac{1}{2\,x-5}\,dx+\int \frac{-\frac{169}{145}+\frac{292}{145}\,x}{x^2+1}\,dx+\int \frac{994}{135}\,\frac{1}{x+2}\,dx+\int -\frac{5}{3}\,\frac{1}{(x+2)^2}\,dx$$

Gli integrali sopra non sono difficili da calcolare:

```
> A := value(%);
```

$$A:=\frac{8023}{1566}\ln(2\,x-5)+\frac{146}{145}\ln(x^2+1)-\frac{169}{145}\arctan(x)+\frac{994}{135}\ln(x+2)+\frac{\frac{5}{3}}{x+2}$$

Naturalmente Maple poteva integrare direttamente:

```
> Int(r/den, x) = int(r/den, x); B := rhs(%):
```

$$\int \frac{29\,x^4+34\,x^3-27\,x^2+4\,x-1}{(2\,x-5)\,(x^2+1)\,(x+2)^2}\,dx=$$

$$\frac{8023}{1566}\ln(2\,x-5)+\frac{146}{145}\ln(x^2+1)-\frac{169}{145}\arctan(x)+\frac{994}{135}\ln(x+2)+\frac{\frac{5}{3}}{x+2}$$

Controlliamo che i risultati siano "uguali" (a meno di una costante additiva):

```
> simplify(A - B);
```

$$0$$

Dunque il risultato ottenuto è corretto.

<u>Nota:</u> si puo ottenere subito la scomposizione in frazione parziale mediante:

```
> r/den = convert(r/den, parfrac, x);
```

$$\frac{29\,x^4+34\,x^3-27\,x^2+4\,x-1}{(2\,x-5)\,(x^2+1)\,(x+2)^2}=\frac{8023}{783}\,\frac{1}{2\,x-5}-\frac{5}{3}\,\frac{1}{(x+2)^2}+\frac{\frac{994}{135}}{x+2}+\frac{\frac{1}{145}(-169+292\,x)}{x^2+1}$$